Comparison of the Jacobian of the marginal one-locus migration-selection equilibrium (E_B) to the mean matrix of the corresponding branching process

Generic model

- **General assumptions and rules**

```
In[1]:= assumeGeneric := {0 ≤ x[1] ≤ 1, 0 ≤ x[2] ≤ 1, 0 ≤ x[3] ≤ 1,
          x[4] == 1 - x[1] - x[2] - x[3], 0 ≤ m ≤ 1, 0 ≤ r ≤ 1 / 2, 0 ≤ qC ≤ 1, 0 < qB < 1}

In[2]:= assumeNoPositionEffect :=
         {w21 → w21, w31 → w13, w32 → w14, w31 → w13, w43 → w34, w42 → w24, w41 → w14, w23 → w14}

In[3]:= wMat := {{w11, w12, w13, w14},
          {w21, w22, w23, w24}, {w31, w32, w33, w34}, {w41, w42, w43, w44}}

In[4]:= simplifyNotation := {x[1] → x1, x[2] → x2, x[3] → x3, x[4] → x4}

In[5]:= w[i_, j_] := wMat[[i, j]]
```

- **Jacobian J for the deterministic two-locus dynamics**

Marginal and mean fitnesses.

```
In[6]:= w[i_] := Sum[w[i, j] x[j], {j, 1, 4}]
        wBar := Sum[w[i] x[i], {i, 1, 4}]
```

Linkage disequilibrium.

```
In[8]:= DD := x[1] x[4] - x[2] x[3]
```

Generic recursion equations for the haplotype frequencies, where q_c is the frequency of allele B_1 on the continent and m is the migration rate.

```
In[9]:= x1RecGenr := (1 - m) (x[1] w[1] - r w[1, 4] DD) / wBar
        x2RecGenr := (1 - m) (x[2] w[2] + r w[1, 4] DD) / wBar
        x3RecGenr := (1 - m) (x[3] w[3] + r w[1, 4] DD) / wBar + m qC
        x4RecGenr := (1 - m) (x[4] w[4] + r w[1, 4] DD) / wBar + m (1 - qC)

In[13]:= recHap := {x1RecGenr, x2RecGenr, x3RecGenr, x4RecGenr}

         recHap /. simplifyNotation // MatrixForm
```

```
(            (1-m) (x1 (w11 x1+w12 x2+w13 x3+w14 x4)-r w14 (-x2 x3+x1 x4))
   ---------------------------------------------------------------------------------------------------------------------------
   x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34 x4)+x4 (w41 x1+w42 x2+w43 x3
             (1-m) (x2 (w21 x1+w22 x2+w23 x3+w24 x4)+r w14 (-x2 x3+x1 x4))
   ---------------------------------------------------------------------------------------------------------------------------
   x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34 x4)+x4 (w41 x1+w42 x2+w43 x3
                   (1-m) (x3 (w31 x1+w32 x2+w33 x3+w34 x4)+r w14 (-x2 x3+x1 x4))
   m qC + ---------------------------------------------------------------------------------------------------------------------
          x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34 x4)+x4 (w41 x1+w42 x2+w4
                   (1-m) (x4 (w41 x1+w42 x2+w43 x3+w44 x4)+r w14 (-x2 x3+x1 x4))
 m (1 - qC) + ---------------------------------------------------------------------------------------------------------------
          x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34 x4)+x4 (w41 x1+w42 x2
```

Generic Jacobian matrix.

In[14]:=
```
JHapGenr := Table[Table[Simplify[D[recHap⟦j⟧, i], Assumptions → assumeGeneric],
    {i, {x[1], x[2], x[3], x[4]}}], {j, {1, 2, 3, 4}}]
JHapGenr /. simplifyNotation // MatrixForm
```

```
(1-m) ((2 w11 x1+w12 x2+w13 x3+(w14-r w14) x4) (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2
-------------------------------------------------------------------------------------------------------------------------------------
                                                                          (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w2
(1-m) ((w21 x2+r w14 x4) (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34
-------------------------------------------------------------------------------------------------------------------------------------
                                                                          (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w2
(1-m) ((w31 x3+r w14 x4) (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34
-------------------------------------------------------------------------------------------------------------------------------------
                                                                          (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w2
(1-m) ((r w14+w41) x4 (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w23 x3+w24 x4)+x3 (w31 x1+w32 x2+w33 x3+w34 x
-------------------------------------------------------------------------------------------------------------------------------------
                                                                          (x1 (w11 x1+w12 x2+w13 x3+w14 x4)+x2 (w21 x1+w22 x2+w2
```

The marginal one-locus migration-selection equilibrium is called E_B and defined as $E_B = (p, q, D) = (0, \hat{q}_B, 0)$, where p and q are the frequencies of A_1 and B_1 on the island, respectively, and D is the linkage disequilibrium. Moreover, $\hat{q}_B$ denotes the equilibrium frequency of B_1. It is defined as the non-trivial solution of the equation obtained by equating the marginal one-locus recursion equation to q. With generic fitnesses, this solution cannot be readily found.

Note that at E_B, the following holds: $\hat{x}_1 = \hat{x}_2 = 0$, $\hat{x}_3 = \hat{q}_B$ and $x_4 = 1 - \hat{q}_B$.

The Jacobian matrix evaluated at the marginal one-locus migration-selection equilibrium E_B.

In[15]:=
```
ruleEB := {x[1] → 0, x[2] → 0, x[3] → qB, x[4] → 1 - qB}
```

Evaluating the Jacobian at the marginal equilibrium E_B:

In[16]:=
```
JHapGenrEB = Simplify[JHapGenr /. ruleEB, Assumptions → assumeGeneric];
JHapGenrEB // MatrixForm
```

```
                           (-1+m) (w14-r w14+qB (w13+(-1+r) w14))
                        - -------------------------------------------------
                           qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44)
                                   (-1+m) (-1+qB) r w14
                          -------------------------------------------------
                           qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44)
(1-m) (-qB (qB (w33-w34)+w34) (w14+qB (w13-w14+w31-w41)+w41)+(r (w14-qB w14)+qB w31) (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w
---------------------------------------------------------------------------------------------------------------------------------
                          (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44))^2
(1-m) (1-qB) (-(w14+qB (w13-w14+w31-w41)+w41) (qB (w43-w44)+w44)+(r w14+w41) (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44)
---------------------------------------------------------------------------------------------------------------------------------
                          (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44))^2
```

As expected (see the case of additive fitnesses above), this is a lower triangular block matrix, $J = \begin{pmatrix} J_1 & 0 \\ J_3 & J_4 \end{pmatrix}$, where

In[17]:=
```
J1 := JHapGenrEB⟦1 ;; 2, 1 ;; 2⟧
J4 := JHapGenrEB⟦3 ;; 4, 3 ;; 4⟧
J1 // FullSimplify // MatrixForm
```

$$\begin{pmatrix} -\frac{(-1+m)\,(qB\,w13+(-1+qB)\,(-1+r)\,w14)}{w44+qB\,(w34+w43-2\,w44+qB\,(w33-w34-w43+w44))} & -\frac{(-1+m)\,qB\,r\,w14}{w44+qB\,(w34+w43-2\,w44+qB\,(w33-w34-w43+w44))} \\ \frac{(-1+m)\,(-1+qB)\,r\,w14}{w44+qB\,(w34+w43-2\,w44+qB\,(w33-w34-w43+w44))} & \frac{(-1+m)\,(-w24+qB\,(r\,w14-w23+w24))}{w44+qB\,(w34+w43-2\,w44+qB\,(w33-w34-w43+w44))} \end{pmatrix}$$

```
J4 // MatrixForm
```

```
  (-1+m) (-1+qB) (2 qB (w33-w34) w44+w34 w44+qB^2 (w33 (w43-2 w44)+w34 w44))      (-1+m) qB (2 qB (w33-w34) w44+w34 w44+qB^2 (w33 (w43-
  ---------------------------------------------------------------------------      ------------------------------------------------------
        (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44))^2                                (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43
   (-1+m) (-1+qB) (2 qB (w33-w34) w44+w34 w44+qB^2 (w33 (w43-2 w44)+w34 w44))       (-1+m) qB (2 qB (w33-w34) w44+w34 w44+qB^2 (w33 (w43
- ---------------------------------------------------------------------------   - ------------------------------------------------------
        (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w43+w44))^2                                (qB (w34+w43-2 w44)+w44+qB^2 (w33-w34-w4
```

■ Mean matrix L for the two-type branching process

Marginal fitnesses of types 1 and 2.

In[19]:=
```
wBP1 := w[1, 3] qB + w[1, 4] (1 - qB)
wBP2 := w[2, 4] (1 - qB) + w[1, 4] qB
```

Mean fitness of resident population.

In[21]:=
```
wBPBar := qB^2 w[3, 3] + 2 qB (1 - qB) w[3, 4] + (1 - qB)^2 w[4, 4]
```

The mean matrix.

```
In[22]:= A := (1 - m) (wBP1 - r (1 - qB) w[1, 4]) / wBPBar
         B := (1 - m) r qB w[1, 4] / wBPBar
         C := (1 - m) r (1 - qB) w[1, 4] / wBPBar
         D := (1 - m) (wBP2 - r qB w[1, 4]) / wBPBar
```

```
In[26]:= L := {{A, C}, {B, D}}
```

```
L // FullSimplify // MatrixForm
```

$$\begin{pmatrix} -\frac{(-1+m)\,(qB\,w13+(-1+qB)\,(-1+r)\,w14)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & \frac{(-1+m)\,(-1+qB)\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \\ -\frac{(-1+m)\,qB\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & \frac{(-1+m)\,(qB\,(-1+r)\,w14+(-1+qB)\,w24)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \end{pmatrix}$$

■ Comparison of L^T and J_1

```
assumeNoPositionEffect
```

{w21 → w21, w31 → w13, w32 → w14, w31 → w13, w43 → w34, w42 → w24, w41 → w14, w23 → w14}

```
Transpose[L] // FullSimplify // MatrixForm
```

$$\begin{pmatrix} -\frac{(-1+m)\,(qB\,w13+(-1+qB)\,(-1+r)\,w14)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & -\frac{(-1+m)\,qB\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \\ \frac{(-1+m)\,(-1+qB)\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & \frac{(-1+m)\,(qB\,(-1+r)\,w14+(-1+qB)\,w24)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \end{pmatrix}$$

```
J1 /. assumeNoPositionEffect // FullSimplify // MatrixForm
```

$$\begin{pmatrix} -\frac{(-1+m)\,(qB\,w13+(-1+qB)\,(-1+r)\,w14)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & -\frac{(-1+m)\,qB\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \\ \frac{(-1+m)\,(-1+qB)\,r\,w14}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} & \frac{(-1+m)\,(qB\,(-1+r)\,w14+(-1+qB)\,w24)}{2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)} \end{pmatrix}$$

```
Transpose[L] - J1 /. assumeNoPositionEffect // Simplify
```

{{0, 0}, {0, 0}}

We note that J_1 is equal to the transpose of the mean matrix L, as long as position and parental effects on relative fitnesses can be ignored. This also holds irrespectively of whether the continent is monomorphic or polymorphic.

■ Eigenvalues of J and L^T

```
FullSimplify[Eigenvalues[J1] /. assumeNoPositionEffect,
 Assumptions → Flatten[{assumeGeneric}]]
```

$$\Big\{-\Big((-1+m)\,(w14-r\,w14+qB\,(w13-w24)+w24)\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)+\sqrt{\Big((-1+m)^2\,\big(((-1+r)\,w14+w24)^2+qB^2\,(w13-2\,w14+w24)\,(w13-2\,w14+4\,r\,w14+w24)-2\,qB\,\big(w14\,((-1+r)\,w13+2\,w14-4\,r\,w14)+(w13+3\,(-1+r)\,w14)\,w24+w24^2\big)\big)\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)^2\Big)}\Big)\Big/\Big(2\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)^2\Big),$$
$$\Big(-(-1+m)\,(w14-r\,w14+qB\,(w13-w24)+w24)\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)+\sqrt{\Big((-1+m)^2\,\big(((-1+r)\,w14+w24)^2+qB^2\,(w13-2\,w14+w24)\,(w13-2\,w14+4\,r\,w14+w24)-2\,qB\,\big(w14\,((-1+r)\,w13+2\,w14-4\,r\,w14)+(w13+3\,(-1+r)\,w14)\,w24+w24^2\big)\big)\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)^2\Big)}\Big)\Big/\Big(2\,\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)^2\Big)\Big\}$$

```
In[27]:= evalsJ1:={-((-1+m) (w14-r w14+qB (w13-w24)+w24) (2 qB (w34-w44)+w44+qB^2 (w33-2 w34+w44)
```

```
In[28]:= evalsJ4 := FullSimplify[Eigenvalues[J4] /. assumeNoPositionEffect,
           Assumptions → Flatten[{assumeGeneric}]]
```

```
evalsJ4
```

$$\left\{0,\ \frac{(-1+m)\,\big(-qB^2\,w33\,w34+(-1+qB)\,(2\,qB\,w33+w34-qB\,w34)\,w44\big)}{\big(2\,qB\,(w34-w44)+w44+qB^2\,(w33-2\,w34+w44)\big)^2}\right\}$$

In[29]:=
```
evalsL := FullSimplify[Eigenvalues[L] /. assumeNoPositionEffect,
  Assumptions → Flatten[{assumeGeneric}]]
```

Check that the eigenvalues of J_1 are the same as those of L:

```
evalsJ1 - evalsL // FullSimplify
```

```
{0, 0}
```

Conclusion

We again use e_1 and e_2 for the eigenvalues of J_1 and e_3 and e_4 for the eigenvalues of J_4.

Conditional on existence of $E_B$as a valid marginal one-locus equilibrium, we would like to know if the condition for invasion of A_1 can be determined exclusively based on the eigenvalues of J_1. If this is the case, we know that what we have shown above for the case of additive fitnesses and a monomorphic continent holds more generally: If invasion of A_1 via E_B is possible in the two-type branching process, then it is also possible in the deterministic two-locus dynamics, and vice versa.

To illustrate the dynamics, consider the 3-simplex Δ^3, which has four verteces, each of which corresponds to the fixation of one out of the four gametes $A_1 B_1$, $A_1 B_2$, $A_2 B_1$ and $A_2 B_2$. Moreover, there are six edges, each of which corresponds to the case where a particular pair of alleles segretages in the population; there are four faces, each of which corresponds to the case where all but one particular allele segregate in the population. The interior of Δ^3 corresponds to all four alleles segregating. Clearly, the marginal one-locus equilibrium E_B sits on the edge that connects the two verteces that correspohd to fixation of $A_2 B_1$ and $A_2 B_2$, respectively. E_B is a valid one-locus polymorphism only if it does not sit on one of these verteces, but on the edge in between.

Matrix J_4 is given by $\begin{pmatrix} \frac{\partial f_3}{\partial x_3} & \frac{\partial f_3}{\partial x_4} \\ \frac{\partial f_4}{\partial x_3} & \frac{\partial f_4}{\partial x_4} \end{pmatrix}$, where $f_i = f_i(x_1, x_2, x_3, x_4)$ is the recursion equation of gamete frequency i and for all $i \in \{1, 2, 3, 4\}$. Therefore, we see that J_4 characterises the dynamics *along* the edge of Δ^3 that connects the verteces $x_4 = 1$ and $x_3 = 1$. From this, it follows that the eigenvalues of J_4 determine the so-called *internal* stability of E_B, that is stability along the edge of Δ^3 on which E_B sits. Matrix J_1, on the other hand, is given by $\begin{pmatrix} \frac{\partial f_1}{\partial x_1} & \frac{\partial f_1}{\partial x_2} \\ \frac{\partial f_2}{\partial x_1} & \frac{\partial f_2}{\partial x_2} \end{pmatrix}$ and therefore characterises the dynamics *transversal* to the boundary of Δ^3 that connects the verteces $x_4 = 1$ and $x_3 = 1$, *i.e.* the dynamics leading into or out of the interior of the simplex. From this, it follows that the eigenvalues of J_1 determine the so-called *external* stability of E_B. Obviously, the external stability is directly linked to the question of wheter or not E_B can be invaded by a mutation at locus A.

As shown above, with generic fitnesses and an arbitrary frequency q_c of B_1 on the continent, one out of the two eigenvalues of J_4 is always 0. We arbitrarily assign 0 to eigenvalue e_3. Then, the value of e_4 determines i) existence of E_B in the one- and two-locus dynamics, and ii) about asymptotic stability of E_B in the one-locus dynamics. Both are required for the initial condition of our biological scenario, and fulfilled if and only if $e_4 < 1$. From this, however, it automatically follows that whenever $E_B$becomes unstable in the two-locus dynamics (upon occurrence of A_1), this can only be due to either e_1 or e_2 being larger than 1. Because e_1 and e_2 are shared between J_1 and L, this argument proof what we wanted to show: If invasion of A_1 via E_B is possible in the two-type branching process, then it is also possible in the deterministic two-locus dynamics, and vice versa.